

matrix utility — List, rename, and drop matrices

[Description](#) [Menu](#) [Syntax](#) [Options](#)
[Remarks and examples](#) [Also see](#)

Description

`matrix dir` lists the names of currently existing matrices. `matrix list` lists the contents of a matrix. `matrix rename` changes the name of a matrix. `matrix drop` eliminates a matrix.

Menu

matrix list

Data > Matrices, ado language > List contents of matrix

matrix rename

Data > Matrices, ado language > Rename matrix

matrix drop

Data > Matrices, ado language > Drop matrices

Syntax

List matrix names

```
matrix dir
```

List contents of matrix

```
matrix list mname [ , noblank nohalf noheader nonames format(%fmt)  
title(string) nodotz ]
```

Rename matrix

```
matrix rename oldname newname
```

Drop matrix

```
matrix drop { _all | mnames }
```

Options

- `noblank` suppresses printing a blank line before printing the matrix. This is useful in programs.
- `nohalf` specifies that, even if the matrix is symmetric, the full matrix be printed. The default is to print only the lower triangle in such cases.
- `noheader` suppresses the display of the matrix name and dimension before the matrix itself. This is useful in programs.
- `nonames` suppresses the display of the bordering names around the matrix.
- `format(%fmt)` specifies the format to be used to display the individual elements of the matrix. The default is `format(%10.0g)`.
- `title(string)` adds the specified title *string* to the header displayed before the matrix itself. If `noheader` is specified, `title()` does nothing because displaying the header is suppressed.
- `nodotz` specifies that `.z` missing values be displayed as blanks.

Remarks and examples

stata.com

▷ Example 1

In the example below, `matrix list` normally displays only the lower half of symmetric matrices. `nohalf` prevents this.

```
. matrix b = (2, 5, 4 \ 5, 8, 6 \ 4, 6, 3)
. matrix a = (1, 2 \ 2, 4)
. matrix dir
      a[2,2]
      b[3,3]
. matrix rename a z
. matrix dir
      z[2,2]
      b[3,3]
. matrix list b
symmetric b[3,3]
      c1  c2  c3
r1   2
r2   5   8
r3   4   6   3
. matrix list b, nohalf
symmetric b[3,3]
      c1  c2  c3
r1   2   5   4
r2   5   8   6
r3   4   6   3
. matrix drop b
. matrix dir
      z[2,2]
. matrix drop _all
. matrix dir
```

□ Technical note

When writing programs and using matrix names obtained through `tempname` (see [P] [macro](#)), it is not necessary to explicitly drop matrices; the matrices are removed automatically at the conclusion of the program.

```
. program define example
1.     tempname a
2.     matrix 'a' = (1,2\3,4)                /* this is temporary */
3.     matrix b = (5,6\7,8)                 /* and this permanent */
4.     display "The temporary matrix a contains"
5.     matrix list 'a', noheader
6. end

. example
The temporary matrix a contains
      c1  c2
r1    1   2
r2    3   4

. matrix dir
      b[2,2]
```

Nevertheless, dropping matrices with temporary names in programs when they are no longer needed is recommended, unless the program is about to exit (when they will be dropped anyway). Matrices consume memory; dropping them frees memory.

□

Also see

[P] [matlist](#) — Display a matrix and control its format

[P] [matrix](#) — Introduction to matrix commands

[U] [14 Matrix expressions](#)

Stata, Stata Press, and Mata are registered trademarks of StataCorp LLC. Stata and Stata Press are registered trademarks with the World Intellectual Property Organization of the United Nations. StataNow and NetCourseNow are trademarks of StataCorp LLC. Other brand and product names are registered trademarks or trademarks of their respective companies. Copyright © 1985–2023 StataCorp LLC, College Station, TX, USA. All rights reserved.



For suggested citations, see the [FAQ on citing Stata documentation](#).